/*
 * Copyright 2017 the original author or authors.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *    http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package geb.module

abstract class NumberLikeInput extends AbstractInput {

    void setNumber(Number number) {
        value(number)
    }

    Number getNumber() {
        parseNumber(value() as String)
    }

    Number getMin() {
        parseNumber(attr("min"))
    }

    Number getMax() {
        parseNumber(attr("max"))
    }

    Number getStep() {
        parseNumber(attr("step"))
    }

    protected Number parseNumber(String string) {
        if (string == null || string.empty) {
            return null
        }
        if (string.bigDecimal) {
            return string.toBigDecimal()
        }
        if (string.integer) {
            return string.toInteger()
        }
        throw new NumberFormatException("Value \"${string}\" of ${this} is not a number.")
    }

}
